IF EXISTS (SELECT * FROM SYSOBJECTS WHERE name='spRpt_PresupuestoAnual' and XTYPE='P')
BEGIN
	DROP PROCEDURE [dbo].[spRpt_PresupuestoAnual]
END
GO
CREATE PROCEDURE [dbo].[spRpt_PresupuestoAnual]
	 @OP VARCHAR(64)=NULL
	,@ParamAuxiliar NVARCHAR(MAX)=NULL
	,@CodPresupuesto_Ini VARCHAR(16)=NULL
	,@CodPresupuesto_Fin VARCHAR(16)=NULL

WITH ENCRYPTION
AS
SET NOCOUNT ON;
BEGIN TRY

/*****     Imprimir    *****/
IF @OP = 'Imprimir'
BEGIN
	SELECT 
		 @CodPresupuesto_Ini=CASE WHEN @CodPresupuesto_Ini IS NULL OR  RTRIM(@CodPresupuesto_Ini)='' THEN REPLICATE(' ',16) ELSE @CodPresupuesto_Ini END
		,@CodPresupuesto_Fin=CASE WHEN @CodPresupuesto_Fin IS NULL OR  RTRIM(@CodPresupuesto_Fin)='' THEN REPLICATE('Z',16) ELSE @CodPresupuesto_Fin END
	
	SELECT
		 t1.[CodPresupuesto] AS [CODIGO PRESUPUESTO]
		,t1.[Vigencia] AS [VIGENCIA]
		,CASE WHEN t1.[Publicado]=1 THEN 'SI' ELSE 'NO' END AS [PUBLICADO]
		,t2.[CodRubro] AS [CODIGO RUBRO]
		,r.NombreRubro AS [NOMBRE RUBRO]
		,t2.[CodCentroCosto] AS [CODIGO CCOSTO]
		,cc.NombreCentroCosto AS [NOMBRE CCOSTO]		
		,t2.ValorActualPresupuesto AS [VALOR ACTUAL]
		,t2.ValorEjecutado AS [VALOR EJECUTADO (COMPROMISOS)]
		,t2.ValorFacturado AS [VALOR EJECUTADO (REGISTROS)]
		,CASE WHEN t2.[Activo]=1 THEN 'SI' ELSE 'NO' END AS [ACTIVO]
	FROM 
		[PresupuestoAnualCabecera] t1
		INNER JOIN [PresupuestoAnualDetalle] t2 ON t1.CodPresupuesto=t2.CodPresupuesto
		INNER JOIN [Rubro] r ON t2.CodRubro=r.CodRubro
		INNER JOIN [RubroTipo] rt ON r.CodTipo=rt.CodTipo
		INNER JOIN [CentroCosto] cc ON t2.CodCentroCosto=cc.CodCentroCosto
	WHERE
		t1.[CodPresupuesto] BETWEEN @CodPresupuesto_Ini AND @CodPresupuesto_Fin
	ORDER BY
		 rt.Orden
		,r.Orden
		
	RETURN 0
END

--- Si se llega a este punto es porque el @OP indicado no Existe en el Procedimiento Almacenado
IF @OP IS NOT NULL
	BEGIN
	SET @ParamAuxiliar='La Operacion "'+@OP+'" No esta definida en el Procedimiento almacenado "'+OBJECT_NAME(@@PROCID)+'", Revise!.'
	RAISERROR(@ParamAuxiliar,16,1)
	RETURN 1
END

--- CONTROLAR Y REGISTRAR LOS ERRORES
END TRY
BEGIN CATCH
	--- Recopilo la informacion del error
	DECLARE @ERROR_NUMBER INT=ERROR_NUMBER(),@ERROR_SEVERITY INT=ERROR_SEVERITY(),@ERROR_STATE INT=ERROR_STATE(),@ERROR_PROCEDURE VARCHAR(128)=ERROR_PROCEDURE(),@ERROR_LINE INT=ERROR_LINE(),@ERROR_MESSAGE VARCHAR(2048)=ERROR_MESSAGE()
	--- Si hay transacciones abiertas, hago un rollback, SE ENCIERRA EN UN (TRY-CATCH) PARA QUE NO SE MUESTRE EL ERROR: Transaction count after EXECUTE indicates a mismatching number of BEGIN and COMMIT
	IF(@@TRANCOUNT>0) BEGIN BEGIN TRY ROLLBACK TRANSACTION END TRY BEGIN CATCH END CATCH END
	--- Ejecuto el SP que controla y registra el error
	EXEC [spSeguridadMessageException] @OP='MostrarMensaje',@ERROR_NUMBER=@ERROR_NUMBER,@ERROR_SEVERITY=@ERROR_SEVERITY,@ERROR_STATE=@ERROR_STATE,@ERROR_PROCEDURE=@ERROR_PROCEDURE,@ERROR_LINE=@ERROR_LINE,@ERROR_MESSAGE=@ERROR_MESSAGE
	RETURN 1
END CATCH;

